

NAG Library Routine Document

F08QFF (DTREXC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08QFF (DTREXC) reorders the Schur factorization of a real general matrix.

2 Specification

```
SUBROUTINE F08QFF (COMPQ, N, T, LDT, Q, LDQ, IFST, ILST, WORK, INFO)
INTEGER          N, LDT, LDQ, IFST, ILST, INFO
REAL (KIND=nag_wp) T(LDT,*), Q(LDQ,*), WORK(N)
CHARACTER(1)    COMPQ
```

The routine may be called by its LAPACK name *dtrexc*.

3 Description

F08QFF (DTREXC) reorders the Schur factorization of a real general matrix $A = QTQ^T$, so that the diagonal element or block of T with row index IFST is moved to row ILST.

The reordered Schur form \tilde{T} is computed by an orthogonal similarity transformation: $\tilde{T} = Z^T T Z$. Optionally the updated matrix \tilde{Q} of Schur vectors is computed as $\tilde{Q} = QZ$, giving $A = \tilde{Q}\tilde{T}\tilde{Q}^T$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: COMPQ – CHARACTER(1) *Input*
On entry: indicates whether the matrix Q of Schur vectors is to be updated.
 COMPQ = 'V'
 The matrix Q of Schur vectors is updated.
 COMPQ = 'N'
 No Schur vectors are updated.
Constraint: COMPQ = 'V' or 'N'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix T .
Constraint: $N \geq 0$.
- 3: T(LDT,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array T must be at least $\max(1, N)$.
On entry: the n by n upper quasi-triangular matrix T in canonical Schur form, as returned by F08PEF (DHSEQR).
On exit: T is overwritten by the updated matrix \tilde{T} . See also Section 9.

- 4: LDT – INTEGER *Input*
On entry: the first dimension of the array T as declared in the (sub)program from which F08QFF (DTREXC) is called.
Constraint: $LDT \geq \max(1, N)$.
- 5: Q(LDQ,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array Q must be at least $\max(1, N)$ if COMPQ = 'V' and at least 1 if COMPQ = 'N'.
On entry: if COMPQ = 'V', Q must contain the n by n orthogonal matrix Q of Schur vectors.
On exit: if COMPQ = 'V', Q contains the updated matrix of Schur vectors.
 If COMPQ = 'N', Q is not referenced.
- 6: LDQ – INTEGER *Input*
On entry: the first dimension of the array Q as declared in the (sub)program from which F08QFF (DTREXC) is called.
Constraints:
 if COMPQ = 'V', $LDQ \geq \max(1, N)$;
 if COMPQ = 'N', $LDQ \geq 1$.
- 7: IFST – INTEGER *Input/Output*
- 8: ILST – INTEGER *Input/Output*
On entry: IFST and ILST must specify the reordering of the diagonal elements or blocks of T . The element or block with row index IFST is moved to row ILST by a sequence of exchanges between adjacent elements or blocks.
On exit: if IFST pointed to the second row of a 2 by 2 block on entry, it is changed to point to the first row. ILST always points to the first row of the block in its final position (which may differ from its input value by ± 1).
Constraint: $1 \leq IFST \leq N$ and $1 \leq ILST \leq N$.
- 9: WORK(N) – REAL (KIND=nag_wp) array *Workspace*
- 10: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

Two adjacent diagonal elements or blocks could not be successfully exchanged. This error can only occur if the exchange involves at least one 2 by 2 block; it implies that the problem is very ill-conditioned, and that the eigenvalues of the two blocks are very close. On exit, T may have been partially reordered, and ILST points to the first row of the current position of the block being moved; Q (if requested) is updated consistently with T .

7 Accuracy

The computed matrix \tilde{T} is exactly similar to a matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

Note that if a 2 by 2 diagonal block is involved in the reordering, its off-diagonal elements are in general changed; the diagonal elements and the eigenvalues of the block are unchanged unless the block is sufficiently ill-conditioned, in which case they may be noticeably altered. It is possible for a 2 by 2 block to break into two 1 by 1 blocks, i.e., for a pair of complex eigenvalues to become purely real. The values of real eigenvalues however are never changed by the reordering.

8 Parallelism and Performance

F08QFF (DTREXC) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $6nr$ if $\text{COMPQ} = \text{'N'}$, and $12nr$ if $\text{COMPQ} = \text{'V'}$, where $r = |\text{IFST} - \text{ILST}|$.

The input matrix T must be in canonical Schur form, as is the output matrix \tilde{T} . This has the following structure.

If all the computed eigenvalues are real, T is upper triangular and its diagonal elements are the eigenvalues.

If some of the computed eigenvalues form complex conjugate pairs, then T has 2 by 2 diagonal blocks. Each diagonal block has the form

$$\begin{pmatrix} t_{ii} & t_{i,i+1} \\ t_{i+1,i} & t_{i+1,i+1} \end{pmatrix} = \begin{pmatrix} \alpha & \beta \\ \gamma & \alpha \end{pmatrix}$$

where $\beta\gamma < 0$. The corresponding eigenvalues are $\alpha \pm \sqrt{\beta\gamma}$.

The complex analogue of this routine is F08QTF (ZTREXC).

10 Example

This example reorders the Schur factorization of the matrix T so that the 2 by 2 block with row index 2 is moved to row 1, where

$$T = \begin{pmatrix} 0.80 & -0.11 & 0.01 & 0.03 \\ 0.00 & -0.10 & 0.25 & 0.35 \\ 0.00 & -0.65 & -0.10 & 0.20 \\ 0.00 & 0.00 & 0.00 & -0.10 \end{pmatrix}.$$

10.1 Program Text

```

Program f08qffe
!      F08QFF Example Program Text
!      Mark 26 Release. NAG Copyright 2016.
!      .. Use Statements ..

```

```

      Use nag_library, Only: dtrexc, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, ifst, ilst, info, ldq,      &
                                ldt, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: q(:,,:), t(:,,:), work(:)
!      .. Executable Statements ..
      Write (nout,*) 'F08QFF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldq = 1
      ldt = n
      Allocate (q(ldq,1),t(ldt,n),work(n))

!      Read T from data file

      Read (nin,*)(t(i,1:n),i=1,n)

!      Read the row indices

      Read (nin,*) ifst, ilst

!      Reorder the Schur factor T

!      The NAG name equivalent of dtrexc is f08qff
      Call dtrexc('No update',n,t,ldt,q,ldq,ifst,ilst,work,info)
      If (info>0) Then
         Write (nout,99999) info, ilst
         Write (nout,*)
         Flush (nout)
      End If

!      Print reordered Schur factor

!      ifail: behaviour on error exit
!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,n,t,ldt,'Reordered Schur form',ifail)

99999 Format (' Reordering could not be completed. INFO = ',I3,' ILST = ',I5)
      End Program f08qffe

```

10.2 Program Data

```

F08QFF Example Program Data
4                               :Value of N
0.80  -0.11  0.01  0.03
0.00  -0.10  0.25  0.35
0.00  -0.65  -0.10  0.20
0.00   0.00   0.00  -0.10   :End of matrix T
2  1                               :Values of IFST and ILST

```

10.3 Program Results

F08QFF Example Program Results

Reordered Schur form

	1	2	3	4
1	-0.1000	-0.6463	0.0874	0.2010
2	0.2514	-0.1000	0.0927	0.3505
3	0.0000	0.0000	0.8000	-0.0117
4	0.0000	0.0000	0.0000	-0.1000
